NUMERICAL INTEGRATION - A MAPLE PERSPECTIVE

JOHN ROUMELIOTIS

Abstract. Peano kernels are most popular for their application in constructing integration rules and (perhaps) more importantly for providing an estimate of the error in the rule. The systematic use of these kernels has produced an abundance of new rules as well as providing many new error bounds for the well known Newton-Cotes quadrature.

In this paper we will outline a simple method that uses the computer algebra system Maple that is able to recapture the well known Ostrowksi, trapezoidal and Simpson's inequalities. Moreover, the technique, which involves manipulation of the Peano kernel, can be adapted to develop new rules, which due to algebraic complexities, have yet to be discovered.

1. Introduction

Newton-Cotes and Newton-Cotes type integration has been an area of much recent activity. Cerone, Dragomir, Pečarić and others have reported on new generalizations of the Ostrowski, trapezoidal and Simpson rules. These results have involved new bounds being expressed in different norms [1, 7, 8, 13, 14] as well as new generalizations of the rules themselves [11, 15]. For a good overview we refer the reader to the survey [1] and the new book [10]. Much may be garnished from the recent work of Cerone [2, 3, 4, 5, 6] whose results are a considerable generalization of previous work.

Most of these results can be obtained via manipulation of the appropriate Peano kernel and nearly all of the algebra required to establish these results require partial integration, differentiation and some simplification - all of which may be done by modern computer algebra systems (CAS's) such as Maple, Mathematics and MuPad. In this paper, we will show how one may prove the Ostrowski and corrected trapezoidal inequalities using Maple [12] and then state two new fourth order integration rules that have been obtained using Maple. A numerical experiment to show the efficiency of these new rules is also undertaken.

2. Ostrowski's inequality

Ostrowski's inequality, a generalized mid-point inequality, is quite simple to establish. The proof involves partial integration of a linear Peano kernel to obtain the rule and integration of the absolute value of the kernel to obtain the bound.

Theorem 1 (Ostrowski). *Let $f : I \subseteq \mathbb{R} \to \mathbb{R}$ be a differentiable mapping in I^o (I^o is the interior of I), and let $a, b \in I^o$ with $a < b$. If $f' : (a, b) \to \mathbb{R}$ is bounded on (a, b), i.e., $\|f'\|_\infty := \sup_{t\in(a,b)} |f'(t)| < \infty$, then we have the inequality:*

$$\left| \int_a^b f(t)\,dt - (b-a)f(x) \right| \le \left[\frac{(b-a)^2}{4} + \left(x - \frac{a+b}{2} \right)^2 \right] \|f'\|_\infty \tag{1}$$

for all $x \in (a, b)$.

Proof. Let the Peano kernel p be given by

$$p(t) = \begin{cases} t-a, & a \le t \le x, \\ t-b, & x < t \le b, \end{cases}$$

for some $x \in [a, b]$. Integrating the product $p(t)f'(t)$ gives

$$\begin{aligned} I &= \int_a^b p(t)f'(t)\,dt && (2)\\ &= \int_a^x p(t)f'(t)\,dt + \int_x^b p(t)f'(t)\,dt && (3)\\ &= (x-a)f(x) - \int_a^x f(t)\,dt - (x-b)f(x) - \int_x^b f(t)\,dt && (4)\\ &= (b-a)f(x) - \int_a^b f(t)\,dt && (5) \end{aligned}$$

Hölder's inequality is used to obtain the bound.

$$\begin{aligned} \left|\int_a^b f(t)\,dt - (b-a)f(x)\right| &= |I| && (6)\\ &= \left|\int_a^b pf'\,dt\right| && (7)\\ &\le \|f'\|_\infty \int_a^b |p|\,dt && (8)\\ &= \|f'\|_\infty \left(\int_a^x t-a\,dt + \int_x^b b-t\,dt\right) && (9)\\ &= \left[\frac{(b-a)^2}{4} + \left(x - \frac{a+b}{2}\right)^2\right] \|f'\|_\infty && (10) \end{aligned}$$

From a computer algebra perspective, the important steps are the integration by parts in (3) to (4) and integration of the positive kernel in (8) to (10). For complex kernels, it is often difficult to ascertain whether a kernel is positive (as in equation (9)), and certainly cannot be automatically determined by a CAS. □

Below, we show the Maple steps used to establish (1). It is helpful to compare each step below with those in (2)–(10) above.

Proof of (1) using Maple.

```
> with( student ):
> p := (t,x) -> t-x;
```

$$p := (t, x) \to t - x$$

```
> Ostr_rule := intparts( Int( p(t,a) * D(f)(t) , t=a..x ) , p(t,a) ) +
> intparts( Int( p(t,b) * D(f)(t) , t=x..b ) , p(t,b) );
```

$$\mathit{Ostr\_rule} := (x-a)\,\mathrm{f}(x) - \int_a^x \mathrm{f}(t)\,dt - (x-b)\,\mathrm{f}(x) - \int_x^b \mathrm{f}(t)\,dt$$

```
> Ostr_rule := collect( combine( Ostr_rule ) , f );
```

$$\mathit{Ostr\_rule} := (b-a)\,\mathrm{f}(x) + \int_a^b -\mathrm{f}(t)\,dt$$

```
> Ostr_bound := int( p(t,a) , t=a..x ) - int( p(t,b) , t=x..b );
```

$$\mathit{Ostr\_bound} := x^2 - \frac{a^2}{2} - a\,(x-a) - \frac{b^2}{2} + b\,(-x+b)$$

```
> Ostr_bound := completesquare( Ostr_bound , x );
```

$$\mathit{Ostr\_bound} := (x - \frac{a}{2} - \frac{b}{2})^2 + \frac{a^2}{4} + \frac{b^2}{4} - \frac{a\,b}{2}$$

```
> Ostr_bound := op(1,Ostr_bound) + factor( subsop(1=0,Ostr_bound) );
```

$$Ostr\_bound := (x - \frac{a}{2} - \frac{b}{2})^2 + \frac{(a-b)^2}{4}$$

□

3. Corrected trapezoidal rule

Theorem 2. *Let $f : I \subseteq \mathbb{R} \to \mathbb{R}$ be a mapping where $f^{(4)}$ exists in I^o (I^o is the interior of I), and let $a, b \in I^o$ with $a < b$. If $f^{(4)} : (a, b) \to \mathbb{R}$ is bounded on (a, b), i.e., $\|f^{(4)}\|_\infty < \infty$, then we have the inequality:*

$$\left| \int_a^b f(t)\,dt - \frac{b-a}{2}\big(f(a) + f(b)\big) + \frac{(b-a)^2}{12}\big(f'(b) - f'(a)\big) \right| \le \frac{(b-a)^5}{720}\|f^{(4)}\|_\infty \tag{11}$$

Since this is a fourth order rule, we will need to integrate a general fourth order polynomial kernel. Doing this will result in all derivatives less than 4 appearing in the expression. We can use a CAS to solve a general non-linear system and find the coefficients of the kernel so that the derivatives vanish. The Maple steps appear below
Proof using Maple.

```
> restart; with(student):
```

Define a general fourth order product Peano kernel

```
> p := t -> 1/4!*(t-A[1])*(t-A[2])*(t-A[3])*(t-A[4]);
```

$$p := t \to \frac{(t - A_1)(t - A_2)(t - A_3)(t - A_4)}{4!}$$

Integrate by parts four times

```
> intparts( Int( p(t) *(D@@4)(f)(t) , t=a..b ) , p(t) );
```

$$\frac{1}{24}(b - A_1)(b - A_2)(b - A_3)(b - A_4)(\mathrm{D}^{(3)})(f)(b)$$
$$- \frac{1}{24}(a - A_1)(a - A_2)(a - A_3)(a - A_4)(\mathrm{D}^{(3)})(f)(a) - \int_a^b ($$
$$\frac{1}{24}(t - A_2)(t - A_3)(t - A_4) + \frac{1}{24}(t - A_1)(t - A_3)(t - A_4)$$
$$+ \frac{1}{24}(t - A_1)(t - A_2)(t - A_4) + \frac{1}{24}(t - A_1)(t - A_2)(t - A_3))(\mathrm{D}^{(3)})(f)(t)dt$$

```
> applyop( intparts , nops(%) , % , diff(p(t),t) ):
> applyop( intparts , nops(%) , % , diff(p(t),t$2) ):
> applyop( intparts , nops(%) , % , diff(p(t),t$3) ):
```

Collect the result in terms of f

```
> Corr_Trap := map( factor , collect( % , f ) );
```

$$Corr\_Trap := \frac{1}{24}(b-A_1)(b-A_2)(b-A_3)(b-A_4)(\mathrm{D}^{(3)})(f)(b)$$
$$-\frac{1}{24}(a-A_1)(a-A_2)(a-A_3)(a-A_4)(\mathrm{D}^{(3)})(f)(a)-\frac{1}{24}(-A_1\,A_3\,A_4-A_2\,A_3\,A_4$$
$$+2\,A_1\,b\,A_4+2\,b\,A_3\,A_4+2\,A_2\,b\,A_4+2\,A_2\,A_3\,b+2\,A_1\,A_3\,b+2\,A_1\,A_2\,b-A_1\,A_2\,A_4$$
$$-A_1\,A_2\,A_3-3\,b^2\,A_4-3\,b^2\,A_3-3\,A_2\,b^2+4\,b^3-3\,A_1\,b^2)(\mathrm{D}^{(2)})(f)(b)+\frac{1}{24}($$
$$-A_1\,A_3\,A_4-A_2\,A_3\,A_4+2\,a\,A_3\,A_4-A_1\,A_2\,A_4-A_1\,A_2\,A_3+2\,A_2\,a\,A_4+2\,A_2\,A_3\,a$$
$$+2\,A_1\,a\,A_4+2\,A_1\,A_3\,a+2\,A_1\,A_2\,a-3\,a^2\,A_4-3\,a^2\,A_3-3\,A_2\,a^2-3\,A_1\,a^2+4\,a^3)$$
$$(\mathrm{D}^{(2)})(f)(a)+\frac{1}{12}(6\,b^2-3\,b\,A_4-3\,b\,A_3+A_3\,A_4-3\,A_2\,b+A_2\,A_4+A_2\,A_3-3\,A_1\,b$$
$$+A_1\,A_4+A_1\,A_3+A_1\,A_2)\mathrm{D}(f)(b)-\frac{1}{12}(6\,a^2-3\,a\,A_4-3\,a\,A_3+A_3\,A_4-3\,A_2\,a$$
$$+A_2\,A_4+A_2\,A_3-3\,A_1\,a+A_1\,A_4+A_1\,A_3+A_1\,A_2)\mathrm{D}(f)(a)$$
$$-\frac{1}{4}(4\,b-A_4-A_3-A_2-A_1)\,\mathrm{f}(b)+\frac{1}{4}(4\,a-A_4-A_3-A_2-A_1)\,\mathrm{f}(a)+\int_a^b \mathrm{f}(t)\,dt$$

Solve for A_1, A_2, A_3, A_4 subject to coefficients of all deriviatives greater than one vanish

```
> eq1 := coeff( Corr_Trap , (D@@2)(f)(a) ) = 0; eq2 := coeff( Corr_Trap
> , (D@@2)(f)(b) )=0; eq3 := coeff( Corr_Trap , (D@@3)(f)(a) ) = 0; eq4
> := coeff( Corr_Trap , (D@@3)(f)(b) ) = 0;
```

$$eq1 := -\frac{1}{24}A_1\,A_3\,A_4-\frac{1}{24}A_2\,A_3\,A_4+\frac{1}{12}a\,A_3\,A_4-\frac{1}{24}A_1\,A_2\,A_4-\frac{1}{24}A_1\,A_2\,A_3+\frac{1}{12}A_2\,a\,A_4$$
$$+\frac{1}{12}A_2\,A_3\,a+\frac{1}{12}A_1\,a\,A_4+\frac{1}{12}A_1\,A_3\,a+\frac{1}{12}A_1\,A_2\,a-\frac{1}{8}a^2\,A_4-\frac{1}{8}a^2\,A_3-\frac{1}{8}A_2\,a^2$$
$$-\frac{1}{8}A_1\,a^2+\frac{a^3}{6}=0$$

$$eq2 := \frac{1}{24}A_1\,A_3\,A_4+\frac{1}{24}A_2\,A_3\,A_4-\frac{1}{12}A_1\,b\,A_4-\frac{1}{12}b\,A_3\,A_4-\frac{1}{12}A_2\,b\,A_4-\frac{1}{12}A_2\,A_3\,b$$
$$-\frac{1}{12}A_1\,A_3\,b-\frac{1}{12}A_1\,A_2\,b+\frac{1}{24}A_1\,A_2\,A_4+\frac{1}{24}A_1\,A_2\,A_3+\frac{1}{8}b^2\,A_4+\frac{1}{8}b^2\,A_3$$
$$+\frac{1}{8}A_2\,b^2-\frac{b^3}{6}+\frac{1}{8}A_1\,b^2=0$$

$$eq3 := -\frac{1}{24}(a-A_1)(a-A_2)(a-A_3)(a-A_4)=0$$

$$eq4 := \frac{1}{24}(b-A_1)(b-A_2)(b-A_3)(b-A_4)=0$$

```
> solve( {eq1,eq2,eq3,eq4} , {A[1],A[2],A[3],A[4]} );
```

$$\{A_4=b,\,A_3=a,\,A_1=b,\,A_2=a\},\,\{A_4=b,\,A_3=a,\,A_2=b,\,A_1=a\},$$
$$\{A_4=b,\,A_2=a,\,A_1=a,\,A_3=b\},\,\{A_1=b,\,A_2=a,\,A_3=b,\,A_4=a\},$$
$$\{A_2=b,\,A_1=a,\,A_3=b,\,A_4=a\},\,\{A_3=a,\,A_1=b,\,A_2=b,\,A_4=a\}$$

Without loss of generality, we can choose the first solution

```
> assign( [%][1] );
```

The corrected trapezoidal rule is

```
> map( factor , Corr_Trap );
```

$$\frac{1}{12}(a-b)^2\,\mathrm{D}(f)(b)-\frac{1}{12}(a-b)^2\,\mathrm{D}(f)(a)+\frac{1}{2}(a-b)\,\mathrm{f}(b)+\frac{1}{2}(a-b)\,\mathrm{f}(a)+\int_a^b \mathrm{f}(t)\,dt$$

The Peano kernel is

```
> p(t);
```

$$\frac{(t-b)^2\,(t-a)^2}{24}$$

Since p is positive, we can merely integrate it to obtain the L1 norm

```
> factor( int( p(t) , t=a..b ) );
```

$$-\frac{(a-b)^5}{720}$$

□

4. New fourth order rules

In this section we state two new fourth order rules that have been derived using the CAS Maple. The first rule is a further correction in the third derivative of the corrected trapezoidal rule, while the other rule is a generalization of Simpson's rule. Both inequalities were derived using a similar technique that that outlined in the last two sections. The Maple worksheets are available from the RGMIA website [16] or on request.

Perturbed corrected trapezoidal inequality

$$\left|\int_a^b f(t)\,dt - \frac{b-a}{2}\big(f(a)+f(b)\big) + \frac{(b-a)^2}{12}\big(f'(b)-f'(a)\big) - \frac{3(b-a)^4}{2048}\big(f'''(b)-f'''(a)\big)\right| \le \frac{5(b-a)^5}{6144}\|f^{(4)}\|_\infty \quad (12)$$

Modified Simpson inequality

$$\left|\int_a^b f(t)\,dt - \frac{b-a}{16}\left(2^{3/4}f(a) + 2\left(2-2^{1/4}\right)\left(\sqrt(2)+2\,2^{1/4}+4\right)f\left(\frac{a+b}{2}\right) + \frac{(b-a)^2}{12}\left(8-3\,2^{3/4}\right)f''\left(\frac{a+b}{2}\right) + 2^{3/4}f(b)\right)\right| \le \frac{2-2^{3/4}}{3840}(b-a)^5\|f^{(4)}\|_\infty \quad (13)$$

To test the efficiency of the rules above, we will integrate

$$\int_0^1 e^{-x^2}\,dx = .7468241328124270254 \quad (14)$$

and compare the result using the trapezoidal rule, corrected trapezoidal rule, perturbed corrected trapezoidal rule, Simpson's rule and the modified Simpson rule.

The results for the trapezoid type rules are shown in Table 1. The perturbed trapezoidal rule is approximately an order of magnitude more accurate than the corrected trapezoidal rule and uses only two more function evaluations. The ratios in the last column show that, as predicted, the rule is of fourth order.

In Table 2, the results for the Simpson rules are shown. For a given interval size, the modified rule far outperforms the generic Simpson rule by over an order of magnitude - this is displayed in the last column. It is important to note that for a given n, the modified rule uses n more function evaluations than Simpson's rule. But even a comparison on this basis is favorable since at 385 function evaluations the modified rule is almost as accurate as Simpson's rule with 513 evaluations.

5. Conclusion

In this paper we outlined a general method that can be used to develop new Newton-Cotes type quadrature rules. The method relies on assuming a general form of the Peano kernel and then using the power of modern computer algebra systems to explore new rules or error bounds.

	Trapezoidal			Corrected Trapezoidal			Perturbed Corrected Trapezoidal		
n	Error		Ratio	Error		Ratio	Error		Ratio
1	0.629E-01	(2)		0.157E-02	(4)		0.584E-03	(6)	
2	0.155E-01	(3)	4.07	0.126E-03	(5)	12.51	0.915E-05	(7)	63.86
4	0.384E-02	(5)	4.02	0.796E-05	(7)	15.78	0.463E-06	(9)	19.78
8	0.959E-03	(9)	4.01	0.499E-06	(11)	15.96	0.277E-07	(13)	16.72
16	0.240E-03	(17)	4.00	0.312E-07	(19)	15.99	0.171E-08	(21)	16.17
32	0.599E-04	(33)	4.00	0.195E-08	(35)	16.00	0.107E-09	(37)	16.04
64	0.150E-04	(65)	4.00	0.122E-09	(67)	16.00	0.666E-11	(69)	16.01
128	0.374E-05	(129)	4.00	0.761E-11	(131)	16.00	0.417E-12	(133)	16.00
256	0.936E-06	(257)	4.00	0.476E-12	(259)	16.01	0.264E-13	(261)	15.76

Table 1. Error in evaluating (14) using the trapezoidal rule, corrected trapezoidal rule (11) and the perturbed corrected trapezoidal rule (12) for a number of intervals n. The number in brackets represents the number of function evaluations.

n	Simpson (*Err1*)		Modified Simpson (*Err2*)		*Err1/Err2*
1	0.356E-03	(3)	0.499E-04	(4)	7.14
2	0.312E-04	(5)	0.248E-05	(7)	12.60
4	0.199E-05	(9)	0.153E-06	(13)	12.96
8	0.125E-06	(17)	0.957E-08	(25)	13.02
16	0.779E-08	(33)	0.598E-09	(49)	13.04
32	0.487E-09	(65)	0.374E-10	(97)	13.04
64	0.305E-10	(129)	0.234E-11	(193)	13.04
128	0.190E-11	(257)	0.146E-12	(385)	13.05
256	0.119E-12	(513)	0.933E-14	(769)	12.73

Table 2. Error in evaluating (14) using Simpson's rule and the modified Simpson rule rule (13) for a number of intervals n. The number in brackets represents the number of function evaluations.

References

[1] N.S. Barnett, P. Cerone, S. S. Dragomir, J. Roumeliotis and A. Sofo, A survey on Ostrowski type inequalities for twice differentiable mappings and applications, *Inequality Theory & Applications*, **1** (2001), Y.J. Cho, S.K. Kim and S.S. Dragomir (Ed.), Nova Sciences Publishers Inc., 35–88.

[2] P. Cerone, Three point rules in numerical integration, *Nonlinear Analysis*, **47** (2001), 2341–2352.

[3] P. Cerone, On perturbed trapezoidal and midpoint rules, *Korean J. of Comput. & Appl. Math.*, **2** (2002), 423-435.

[4] P. Cerone, On an identity for the Chebychev functional and some ramifications, *J. Inequal. Pure & Appl. Math.*, **3**(1) (2002), Article 4. [URL] `http://jipam.vu.edu.au/v3n1/034_01.html`

[5] P. Cerone, Generalised trapezoidal rules with error involving bounds of the nth derivative, *Math. Ineq. & Appl.*, **5**(3) (2002), 451–462.

[6] P. Cerone, On relationships between Ostrowski, trapezoidal and Chebychev identities and inequalities, accepted for publication in *Soochow J. Math.* A preprint is available from the *RGMIA Research Report Collection*, **4**(2) (2001), Article 14. [URL] `http://rgmia.vu.edu.au/v4n2.html`

[7] S.S. Dragomir, On Simpson's quadrature formula for differentiable mappings whose derivatives belong to L_p-spaces and applications, *J. KSIAM*, **2**(2) (1998), 57–65.

[8] S.S. Dragomir, On Simpson's quadrature formula for mappings of bounded variation and applications, *Tamkang J. Math.*, **30**(1) (1999), 53–58.

[9] S.S. Dragomir, R.P. Agarwal and P. Cerone, On Simpson's inequality and applications, *J. of Inequal. & Appl.*, **5** (2000), 533–579.

[10] S.S. Dragomir and Th.M. Rassias (eds.), *Ostrowski Type Inequalities and Applications in Numerical Integration*, Kluwer Academic, Dordrecht, 2002.

[11] S.S. Dragomir, J. Pečarić and S. Wang, The unified treatment of trapezoid, Simpson, and Ostrowski type inequality for monotonic mappings and applications, *Math. Comput. Modelling*, **31** (2000), 61–70.

[12] M.B. Monagan, K.O. Geddes, K.M. Heal, G. Labahn, S.M. Vorkoetter, J. McCarron and P. DeMarco, *Maple 8 Introductory Programming Guide*, Waterloo Maple Inc., Waterloo, 2002.
[13] J. Pečarić and S. Varošanec, A note on Simpson's inequality for functions of bounded variation, *Tamkang J. Math.*, **31**(3) (2000), 239–242.
[14] J. Pečarić and S. Varošanec, Simpson's formula for functions whose derivatives belong to L_p spaces, *Appl. Math. Lett.*, **14**(2), (2001), 131–135.
[15] J. Pečarić and S. Varošanec, Harmonic polynomials and generalization of Ostrowksi inequality with applications in numerical integration, *Nonlinear Analysis*, **47** (2001), 2365–2374.
[16] RGMIA Web site (Maple section): `http://rgmia.vu.edu.au/maple/`.
[17] G.-S. Yang and H.-F. Chu, A note on Simpson's inequality for function of bounded variation, *Tamsui Oxf. J. Math. Sci*, **16**(2) (2000), 229–240.

School of Communications and Informatics, Victoria University, PO Box 14428, Melbourne City MC 8001, AUSTRALIA
E-mail address: `John.Roumeliotis@vu.edu.au`
URL: `http://www.staff.vu.edu.au/johnr`